
PROBLEM: 
A fragile glass sheet is hit by a an explosive blast wave and fails, generating a cloud 
of fragments. 
MESH: 
The sheet is meshed by Q4GS 4-node shell elements. The blast pressure wave is 
applied to the sheet by means of CL3Q boundary condition elements. A PMAT 
element is used to visualize the initial position of the explosive charge, and 4 FUN3 
elements are used to visualize the external frame of the sheet, which is held fixed. 
GEOMETRIC COMPLEMENTS 
Debris elements (DEBR) are associated with the whole sheet. By using PLEV 0, just 
one spherical particle replaces each failing shell element. 
MATERIALS: 
The glass sheet uses a linear elastic material (VM23) with an associated failure 
criterion (FAIL PEPR). A special impedance material IMPE AIRB is associated with 
the CL3Q elements that are used to apply the blast pressure wave to the sheet. A 
phantom (FANT) material is associated with the FUN3 elements used to visualize the 
frame. 
BOUNDARY CONDITIONS: 
The sheet is clamped all around its perimeter. 
LOADING: 
AIRB pressure loading is applied. 
CALCULATIONS: 
A calculation is performed until a physical time of 3 ms, when the sheet is completely 
failed. 
RESULTS: 
The sheet fragments as expected. Some macro fragments survive until the end of the 
simulation, together with a cloud of DEBR particles. 

POST-TREATMENT 
Animations are produced. 
Numerical Solutions 
TEST23 
The Cast3m mesh generation file reads: 
 
OPTI echo 1; 
opti titr 'test23'; 
OPTI sauv form 'test23.msh'; 
opti trac psc ftra 'test23.ps'; 
OPTI dime 3 elem qua4; 
den=0.02; 
DENS den; 
sizx = 1.00; 
sizy = 1.40; 
p0 = 0 0 0; 
p1 = sizx 0 0; 
p2 = sizx sizy 0; 
p3 = 0 sizy 0; 
pairb = 0.5 0.7 -3.5; 
glas = (dall (p0 d p1) (p1 d p2) (p2 d p3) (p3 d p0) plan) coul bleu; 
pres = glas coul roug; 
!pres = (dall (p0 d p3) (p3 d p2) (p2 d p1) (p1 d p0) plan) coul bleu; 
e0 = pres elem cont p0; 
e1 = pres elem cont p1; 
e2 = pres elem cont p2; 

e3 = pres elem cont p3; 
tple = e0 et e1 et e2 et e3; 
lbq = p0 d p1 d p2 d p3 d; 
fram = p0 d 1 p1 d 1 p2 d 1 p3 d 1 p0; 
bubb = (manu poi1 pairb) coul roug; 
mesh = glas et lbq et pres et bubb et tple et fram; 
elim mesh; 
TASS mesh; 
sauv form mesh; 
trac qual mesh; 
* 
list(nbno glas); 
list(nbel glas); 
list(nbno pres); 
list(nbel pres); 
list(nbno bubb); 
list(nbel bubb); 
list(nbno mesh); 
list(nbel mesh); 
fin;$ 

 

The EUROPLEXUS input file reads: 
 
TEST23 
ECHO 
!CONV win 
TRID LAGR FAIL 0.0 
CAST FORM mesh 
DIME 
  PT6L 3622 PT3L 3500 Q4GS 3500 CL3Q 3500 PMAT 1 FUN3 4 DEBR 3500 ZONE 5 
  FORC 100 
  TABL 1 3 
  PRES 33 26 
TERM 
GEOM  
  Q4GS glas  
  PMAT bubb 
  FUN3 fram 
  CL3Q pres 
TERM 
COMP EPAI 1.E-4 LECT glas TERM 
          1.E-1 LECT bubb TERM 
          1.E-4 LECT fram TERM 
     DEBR ROF 1.0  
          FILL PLEV 0 

               RO 2500. DRAG 1.D0 TRAJ OBJE LECT glas TERM 
     GROU 1 'part' LECT 7006 PAS 1 10505 TERM 
     COUL roug LECT part  TERM 
MATE VM23 RO 2500 YOUN 7E10 NU 0.23 ELAS 1.756E10 
          FAIL PEPR LIMI 0.1 
          TRAC 2 1.756E10 0.15 1.8E10 1.15 
          LECT glas TERM 
!    IMPE AIRB X 0.5 Y 0.7 Z -3.5 MASS 3 TINT -2.E-3 
     IMPE AIRB NODE LECT bubb TERM MASS 3 TAUT 
          LECT pres TERM 
     MASS 1.0 LECT bubb TERM 
     FANT 1.0 LECT fram TERM 
LINK COUP FREQ 1 
     BLOQ 123456 LECT lbq TERM 
ECRI DEPL VITE TFRE 3.E-3 
     FICH ALIC TEMP FREQ 1 
          ELEM LECT tple TERM 
     FICH SPLI ALIC TFRE 3.D-5 
OPTI NOTE CSTA 0.5 LOG 1  
CALC TINI 0. TEND 3.D-3 
Fin 

 
The EUROPLEXUS post-treatment file reads: 
 
Post-treatment 
ECHO 
* 
RESU SPLI ALIC 'test23.ali' GARD PSCR 
COMP GROU 1 'part' LECT 7006 PAS 1 10505 TERM 
* 
SORT VISU NSTO 1 
*================================================================= 
PLAY 
CAME   1 EYE  -2.51740E+00 -1.05337E+00  2.51415E+00 
!        Q     8.50710E-01  3.34786E-01 -3.09280E-01 -2.61834E-01 
         VIEW  7.01532E-01  4.07651E-01 -5.84529E-01 
         RIGH  6.71577E-01 -6.52575E-01  3.50899E-01 
         UP    2.38404E-01  6.38723E-01  7.31571E-01 
         FOV   2.48819E+01 
CAME   2 EYE   1.35836E+00 -3.40898E+00  2.68476E+00 
!        Q     8.56960E-01  3.69966E-01  2.05747E-01 -2.93961E-01 
         VIEW -1.35123E-01  7.55055E-01 -6.41587E-01 
         RIGH  7.42511E-01 -3.51586E-01 -5.70145E-01 
         UP    6.56064E-01  5.53425E-01  5.13129E-01 
         FOV   2.48819E+01 
SCEN GEOM NAVI FREE 
          FACE SBAC 
          POIN SPHP 
     ISO  FILL FIEL DEPL SCAL USER PROG 0.5E-2 PAS 0.5E-2 7.E-2 TERM 
     VECT SCCO FIEL VITE SCAL USER PROG 40. PAS 40. 560. TERM 
     TEXT ISCA 
     TEXT VSCA 
     COLO PAPE 

     LIMA ON 
SLER CAM1 1 NFRA 1 
FREQ 1 
TRAC OFFS FICH AVI NOCL NFTO 101 FPS 10 KFRE 10 COMP -1 
  OBJE NFAI LECT glas part fram TERM REND 
GOTR LOOP 99 OFFS FICH AVI CONT NOCL 
  OBJE NFAI LECT glas part fram TERM REND 
GO 
TRAC OFFS FICH AVI CONT 
  OBJE NFAI LECT glas part fram TERM REND 
ENDPLAY 
*================================================================= 
SUIT 
Post-treatment (time curves from alice temps file) 
ECHO 
* 
RESU ALIC TEMP 'test23.alt' GARD PSCR 
* 
SORT GRAP 
* 
AXTE 1.0 'Time [s]'  
* 
COUR 1 'p_e0' ECRO COMP 1 ELEM LECT e0 TERM 
COUR 2 'p_e1' ECRO COMP 1 ELEM LECT e1 TERM 
COUR 3 'p_e2' ECRO COMP 1 ELEM LECT e2 TERM 
COUR 4 'p_e3' ECRO COMP 1 ELEM LECT e3 TERM 
TRAC 1 2 3 4 AXES 1.0 'Pressure [Pa]' 
FIN 

 

The resulting AIRB pressure at the four corners of the sheet is: 

The fragmentation process and the velocities are shown in the next figures: 

Formation of secondary debris 
 

 
 
A glass sheet is loaded by an air blast wave and breaks up into fragments (debris 
particles). These particles impact a second glass sheet nearby and break it up into 
secondary debris particles. 
 
TEST27 
 
The Cast3m mesh generation file reads: 
 
OPTI echo 1; 
opti titr 'test27'; 
OPTI sauv form 'test27.msh'; 
opti trac psc ftra 'test27_mesh.ps'; 
OPTI dime 3 elem qua4; 
den=0.10; 
DENS den; 
sizx = 1.00; 
sizy = 1.40; 
p0 = 0 0 0; 
p1 = sizx 0 0; 
p2 = sizx sizy 0; 
p3 = 0 sizy 0; 
pairb = 0.5 0.7 -3.5; 
glas = (dall (p0 d p1) (p1 d p2) (p2 d p3) (p3 d p0) plan) coul roug; 
glas2 = (glas plus (0 0 0.5)) coul vert; 
pres = glas coul jaun; 
e0 = pres elem cont p0; 
e1 = pres elem cont p1; 
e2 = pres elem cont p2; 
e3 = pres elem cont p3; 
tple = e0 et e1 et e2 et e3; 

lbq = p0 d p1 d p2 d p3 d; 
lbq = lbq et (cont glas2); 
fram = p0 d 1 p1 d 1 p2 d 1 p3 d 1 p0; 
fram = fram et (fram plus (0 0 0.5)); 
bubb = (manu poi1 pairb) coul roug; 
mesh = glas et glas2 et lbq et pres et bubb et tple et fram; 
elim mesh; 
TASS mesh; 
sauv form mesh; 
trac qual mesh; 
* 
list(nbno glas); 
list(nbel glas); 
list(nbno glas2); 
list(nbel glas2); 
list(nbno pres); 
list(nbel pres); 
list(nbno bubb); 
list(nbel bubb); 
list(nbno mesh); 
list(nbel mesh); 
fin; 

 

The EUROPLEXUS input file reads: 
 
TEST27 
ECHO 
 CONV win 
TRID LAGR EROS 0.0 
CAST FORM mesh 
DIME 
  PT6L 330 PT3L 289 Q4GS 280 CL3Q 140 PMAT 1 FUN3 8 DEBR 280 ZONE 5 
TERM 
GEOM  
  Q4GS glas glas2  
  PMAT bubb 
  FUN3 fram 
  CL3Q pres 
TERM 
COMP EPAI 1.E-4 LECT glas glas2 TERM 
          1.E-1 LECT bubb TERM 
          1.E-4 LECT fram TERM 
     DEBR ROF 1.0  
          FILL PLEV 0 
               RO 2500. DRAG 1.D0 TRAJ OBJE LECT glas TERM 

          FILL PLEV 0 
               RO 2500. DRAG 1.D0 TRAJ OBJE LECT glas2 TERM 
     GROU 3 'part1' LECT 430 PAS 1 569 TERM 
            'part2' LECT 570 PAS 1 709 TERM 
            'part'  LECT part1 part2   TERM 
     COUL roug LECT part1 TERM 
          vert LECT part2 TERM 
MATE VM23 RO 2500 YOUN 7E10 NU 0.23 ELAS 1.756E10 
         !FAIL PEPR LIMI 0.1 
          FAIL PEPR LIMI 0.05 
          TRAC 2 1.756E10 0.15 1.8E10 1.15 
          LECT glas glas2 TERM 
     IMPE AIRB NODE LECT bubb TERM MASS 3 TAUT 
          LECT pres TERM 
     MASS 1.0 LECT bubb TERM 
     FANT 1.0 LECT fram TERM 
LINK COUP FREQ 1 
     BLOQ 123456 LECT lbq TERM 
     PINB BODY        DMIN 0.02 LECT glas2 TERM 
          BODY MLEV 0 DIAM 0.02 LECT part1 TERM 

ECRI DEPL VITE TFRE 3.E-3 
     FICH ALIC FREQ 1 
OPTI NOTE CSTA 0.5 LOG 1  
CALC TINI 0. TEND 3.D-3 
*================================================================= 
PLAY 
CAME   1 EYE  -3.25928E+00  1.61671E-01 -1.27078E+00 ! From below 
!        Q     5.00268E-01  4.44242E-01 -7.09366E-01 -2.21768E-01 
         VIEW  9.06783E-01  1.29851E-01  4.01102E-01 
         RIGH -1.04761E-01 -8.52147E-01  5.12708E-01 
         UP   -4.08374E-01  5.06935E-01  7.59109E-01 
         FOV   2.48819E+01 
SCEN GEOM NAVI FREE 
          FACE SBAC 
          LINE HEOU SFRE 
          POIN SPHP FACT 2.0 
          PINB CDES 
     VECT SCCO FIEL VITE SCAL USER PROG 40. PAS 40. 560. TERM 

     TEXT VSCA 
     COLO PAPE 
     LIMA ON 
          LIGY -1 
         !SELE BRAS 
         !APPL MESH 
SLER CAM1 1 NFRA 1 
FREQ 1 
TRAC OFFS FICH AVI NOCL NFTO 331 FPS 25 KFRE 10 COMP -1 
     OBJE NFAI LECT glas glas2 part fram TERM REND 
GOTR LOOP 329 OFFS FICH AVI CONT NOCL 
     OBJE NFAI LECT glas glas2 part fram TERM REND 
GO 
TRAC OFFS FICH AVI CONT 
     OBJE NFAI LECT glas glas2 part fram TERM REND 
ENDPLAY 
*================================================================= 
FIN 

 

The fragmentation process is shown in the next figure: 

Crack formation in Laminated Glass 
 

 
 
A laminated glass sheet is loaded by an air blast wave. After va certain time, cracks 
are formed whose pattern resembles the experimentally observed one. 
 
EMI_LS25_DKT3 
 
The Cast3m mesh generation file reads: 
 
OPTI echo 1; 
OPTI dime 3 elem tet4; 
den=0.0125; 
DENS den; 
sizx = 1.1; 
sizy = 0.9; 
rand = 0.05; 
b11 = rand rand 0; 
b12 = (sizx-rand) rand 0; 
b13 = (sizx-rand) (sizy-rand) 0; 
b14 = rand (sizy-rand) 0; 
li1 = b11 d b12 d b13 d b14 d; 
a_airb = surf li1 plan; 
 
b31 = 0 0 0; 
b32 = sizx 0 0; 
li3 = b31 d b32 d b12 d b11 d; 
a3 = surf li3 plan; 
 
b41 = sizx sizy 0; 
li4 = b32 d b41 d b13 d b12 d; 
a4 = surf li4 plan; 
 
b51 = 0 sizy 0; 
li5 = b41 d b51 d b14 d b13 d; 
a5 = surf li5 plan; 
 
li6 = b31 d b11 d b14 d b51 d; 
a6 = surf li6 plan; 
 
a_rand = a3 et a4 et a5 et a6; 
a_glass1 = a_airb et a_rand; 
 
p1 = 0 0 0.004; 
p2 = 0 0 -0.004; 
v1 = a_rand volu 'TRAN' p1; 

v1 = v1 et (a_rand volu 'TRAN' p2); 
 
pch1 = ((sizx+rand)/2.) ((sizy+rand)/2.) 1.8; 
ch1 = manu poi1 pch1; 
 
pcent = ((sizx+rand)/2.) ((sizy+rand)/2.) 0; 
 
pdis = a_glass1 poin 'PROC' pcent; 
 
edis1 = a_glass1 ELEM CONTENANT pdis; 
 
a_v1 = v1 enve; 
 
kod1 = faux; 
REPE I0 (NBEL (a_v1 ELEM TRI3)); 
  xx yy zz = coor (bary (a_v1 ELEM TRI3 &I0)); 
  SI ( zz ega 0.004 0.001); 
     si (kod1);bound=bound et (a_v1 ELEM TRI3 &I0); 
     sinon; bound = a_v1 ELEM TRI3 &I0;kod1 = vrai; 
     fins; 
  FINS; 
  SI ( zz ega -0.004 0.001); 
     si (kod1);bound=bound et (a_v1 ELEM TRI3 &I0); 
     sinon; bound = a_v1 ELEM TRI3 &I0;kod1 = vrai; 
     fins; 
  FINS; 
FIN I0; 
 
 
modell = v1 et a_glass1 et a_airb et pdis et  
         edis1 et ch1 et bound; 
elim modell; 
 
TASS modell; 
OPTI sauv form 'emi_ls25_dkt3.msh'; 
sauv form modell; 

 

The EUROPLEXUS input file reads: 
 
emi_ls25 with model lsgl 
$ 
 ECHO 
! VERI 
! CONV WIN 
CAST modell 
TRID LAGR EROS 1.0 
OPTI TOLC 1e-1 
$ 

DIME 
  PT6L       12000     ZONE  4 
  PR6         5708     CL3T        14808     DKT3  14808 
  PMAT 1 
  ECRO     3373653 
  LIAI       10400 
  BLOQ       10400 
  MNTI 102 
  NGPZ 5 

TERM 
$ 
GEOM 
  PR6 v1 
  DKT3 a_glass1 
  PMAT ch1 
  CL3T a_airb 
TERM 
$ 
COMP 
 EPAI 0.0075 LECT a_glass1 TERM 
      0.05 LECT ch1 TERM ! only for visualization 
 SAND    3 
  FRAC 0.4 0.2 0.4 
  NGPZ   2 1 2  
  LECT a_glass1 term 
 COUL roug LECT ch1 TERM 
      turq LECT a_glass1 TERM 
      jaun LECT a_airb TERM 
      vert LECT v1 TERM 
MATE 
      LSGL RO 2500 YOUN 7E10 NU 0.23 CORR 16.0 
           FAIL PEPR LIMI 0.0012 
           LECT a_glass1 TERM 
           laye lect 1 3 term 
      VM23 RO 1100. YOUNG 3E6 NU 0.46 ELAS 3.45E9 
           TRAC 1 3.45E9 15 
           LECT a_glass1 TERM 
           laye lect 2 term 
      VM23 RO 2770. YOUNG 3.5e6 NU 0.42 ELAS 3.45E9 
           TRAC 1 3.45E9 0.15 
           LECT v1 TERM 
      IMPE AIRB NODE LECT ch1 TERM CONF 1 MASS 0.09 TAUT 
           LECT a_airb TERM 
      FANT 1.E-3 LECT ch1 TERM 
LIAI BLOQ 123 bound 
  
ECRI DEPL VITE TFRE 5.E-3 
     FICH SPLI ALIC tfreq 1e-4  
     FICH ALIC TEMP tfreq 5e-5 
               POIN LECT pdis TERM 
               ELEM LECT 4148 1628 1698 edis1 TERM 
$ 
OPTI NOTE LOG 1 
$ 
CALC TINI 0 TEND 10e-3 
*================================================================= 
SUIT 
Post-treatment (time curves from alice temps file) 
ECHO 
* 
RESU ALIC TEMP GARD PSCR 
SORT GRAP 
AXTE 1.0 'Time [s]'  
COUR 1 'sig_edis' CONT COMP 1 GAUZ 0 ELEM LECT 4148 TERM 
COUR 2 'sig_edis' CONT COMP 2 GAUZ 0 ELEM LECT 4148 TERM 
COUR 3 'sig_edis' ECRO COMP 12 GAUZ 0 ELEM LECT 4148 TERM 
COUR 4 'sig_edis' ECRO COMP 9 GAUZ 0 ELEM LECT 4148 TERM 
COUR 5 'sig_edis' ECRO COMP 4 GAUZ 0 ELEM LECT 4148 TERM 
COUR 6 'sig_edis' ECRO COMP 5 GAUZ 0 ELEM LECT 4148 TERM 
COUR 7 'sig_edis' ECRO COMP 8 GAUZ 0 ELEM LECT 4148 TERM 

COUR 11 'sig_edis' CONT COMP 1 GAUZ 0 ELEM LECT 1628 TERM 
COUR 12 'sig_edis' CONT COMP 2 GAUZ 0 ELEM LECT 1628 TERM 
COUR 13 'sig_edis' ECRO COMP 12 GAUZ 0 ELEM LECT 1628 TERM 
COUR 14 'sig_edis' ECRO COMP 9 GAUZ 0 ELEM LECT 1628 TERM 
COUR 15 'sig_edis' ECRO COMP 4 GAUZ 0 ELEM LECT 1628 TERM 
COUR 16 'sig_edis' ECRO COMP 5 GAUZ 0 ELEM LECT 1628 TERM 
COUR 17 'sig_edis' ECRO COMP 8 GAUZ 0 ELEM LECT 1628 TERM 
COUR 21 'sig_edis' CONT COMP 1 GAUZ 0 ELEM LECT 1698 TERM 
COUR 22 'sig_edis' CONT COMP 2 GAUZ 0 ELEM LECT 1698 TERM 
COUR 23 'sig_edis' ECRO COMP 12 GAUZ 0 ELEM LECT 1698 TERM 
COUR 24 'sig_edis' ECRO COMP 9 GAUZ 0 ELEM LECT 1698 TERM 
COUR 25 'sig_edis' ECRO COMP 4 GAUZ 0 ELEM LECT 1698 TERM 
COUR 26 'sig_edis' ECRO COMP 5 GAUZ 0 ELEM LECT 1698 TERM 
COUR 27 'sig_edis' ECRO COMP 8 GAUZ 0 ELEM LECT 1698 TERM 
COUR 9 SOMME 3 5 1 6 1 7 1 
COUR 19 SOMME 3 15 1 16 1 17 1 
COUR 29 SOMME 3 25 1 26 1 27 1 
COUR 99 'dz_pdis' DEPL COMP 3 NOEU LECT pdis TERM 
trac 1 2  axes 1.0 'SIG' 
trac 9  axes 1.0 'EPSHYD' 
trac 3 4  axes 1.0 'FAIL' 
trac 11 12  axes 1.0 'SIG' 
trac 19  axes 1.0 'EPSHYD' 
trac 13 14  axes 1.0 'FAIL' 
trac 21 22  axes 1.0 'SIG' 
trac 29  axes 1.0 'EPSHYD' 
trac 23 24  axes 1.0 'FAIL' 
trac 99 text axes 1.0 'DISPL. [M]' 
SUIT 
*================================================================= 
emi12_avi 
 ECHO 
 CONV WIN 
RESU SPLI ALIC GARD PSCR 
* 
OPTI PRIN 
* 
SORT VISU NSTO 1 
*================================================================= 
PLAY 
CAME   1 EYE   5.00000E-01  4.00000E-01  6.42268E+00 
!        Q     1.00000E+00  0.00000E+00  0.00000E+00  0.00000E+00 
         VIEW  0.00000E+00  0.00000E+00 -1.00000E+00 
         RIGH  1.00000E+00  0.00000E+00  0.00000E+00 
         UP    0.00000E+00  1.00000E+00  0.00000E+00 
         FOV   1.08819E+01 
SCEN GEOM NAVI FREE 
          FACE SBAC 
          POIN SPHP 
     ISO  FILL FIEL ECRO 12 SCAL A14 
     TEXT ISCA 
     COLO PAPE 
sler cam1 1        nfra 1 
freq 200 
go 
trac offs fich bmp 
     obje lect a_glass1 term   rend 
ENDPLAY 
*================================================================= 
FIN 

 

The cracking process is shown in the next figure: 
 

   

   
 


